
Subject: Re: calculation of exclusive breastfeeding
Posted by [sumonrupop](#) on Sun, 10 Jul 2016 16:51:59 GMT
[View Forum Message](#) <> [Reply to Message](#)

Hi Trevor,

I used the code you given to calculate exclusive breastfeeding by using Bangladesh Demographic and Health survey 2014 data. I used the KR file. However, by using these code I found only 13.9% exclusive breastfeeding. In BDHS 2014 report this percentage is 55%. Can you help me to calculate the percentage reported in BDHS report?. Here the code i used .

use BDKR70FL.DTA

* Last child in the last 2 years living with mother

* age in months

gen age = v008-b3

* drop if too old or not alive

keep if age<24 & b5==1

* recode age into groups

recode age (0/1=1 "0-1")(2/3=2 "2-3")(4/5=3 "4-5")(6/8=4 "6-8")(9/11=5 "9-11") ///
(12/17=6 "12-17")(18/23=7 "18-23")(24/59=.) , gen(child_age)

* tab of all living children born in the last 2 years

tab child_age

tab child_age [iw=v005/1000000]

* keep only those children living with mother ...

keep if b9==0

* ... and keep the last born of those

drop if _n > 0 & caseid == caseid[_n-1]

* check the deonimnator

tab child_age

tab child_age [iw=v005/1000000]

* breastfeeding status.

gen water=0

gen liquids=0

gen milk=0

gen solids=0

gen breast=0

* Water

replace water=1 if (v409>=1 & v409<=7)

* Other non-milk liquids

* check for country specific liquids

foreach xvar of varlist v409a v410 v410a v412c v413* {

replace liquids=1 if `xvar'>=1 & `xvar'<=7

}

```
* Powdered or tinned milk, formula, fresh milk
foreach xvar of varlist v411 v411a v412 {
replace milk=1 if `xvar'>=1 & `xvar'<=7
}
```

```
* Solid food
* check for country specific foods
foreach xvar of varlist v412a v412b v414* {
replace solids=1 if `xvar'>=1 & `xvar'<=7
}
```

```
* Still breastfeeding
replace breast=1 if m4==95
```

```
tab1 water liquids milk solids breast
```

```
* Generate column variable used in table 11.3
gen feeding=1
replace feeding=2 if water==1
replace feeding=3 if liquids==1
replace feeding=4 if milk==1
replace feeding=5 if solids==1
replace feeding=0 if breast==0
tab feeding,m
```

```
label define feeding 0 "Not breastfeeding" 1 "exclusive breastfeeding" 2 "+Water" 3 "+Liquids" 4
"+Other Milk" 5 "+Solids"
label val feeding feeding
```

```
tab child_age feeding [iweight=v005/1000000], row
```

Thanks in advance

Nuruzzaman
